# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=deepin-application-manager
pkgver=1.0.20
pkgrel=1
pkgdesc='App manager of Deepin Desktop Environment'
arch=('loong64' 'x86_64')
url="https://github.com/linuxdeepin/dde-application-manager"
license=('GPL3')
# startdde: org.deepin.dde.XSettings1
depends=('qt5-base' 'dtkcore' 'dtkwidget' 'xcb-util-wm' 'glibc' 'libxcb' 'libxres' 'gcc-libs'
         'libx11' 'glib2' 'gsettings-qt' 'startdde')
makedepends=('cmake' 'ninja')
groups=('deepin')
source=("https://github.com/linuxdeepin/dde-application-manager/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha512sums=('d7ce6a1fbcad66fecf5a98b8008334c7330ff8190c609fa1c6fa42ddfa693402c1bd84b07c27ea2becc98f499f4ab8698802b670252b7188586d67c8fb0fcd79')

build() {
  cd dde-application-manager-$pkgver
  cmake . -GNinja -DCMAKE_INSTALL_PREFIX=/usr -DCMAKE_INSTALL_SYSCONFDIR=/etc
  ninja
}

package() {
  cd dde-application-manager-$pkgver
  DESTDIR="$pkgdir" ninja install
}
